
MetaQuotes Language 4 

 

Пример за създаване на експертна система 

Принципите за създаване на потребителски програми на езика MQL 4 са разгледани в примера за 
създаване на най-проста експертна система въз основа на стандартния индикатор MACD. В тази 
експертна система се разглеждат и примери за осъществяване на такива функции като: определяне 
на нивата на тейкпрофит, поставяне на плаващ стоп, както и създаване на условия за безопасна 
работа. В този конкретен пример за създаване на експертна система търговията се извършва въз 
основа само на една отворена позиция.

Принципи на търговията:

• влизане в Long (BUY) – индикаторът MACD е под нулата, движи се отдолу нагоре,

сигналната линия го пресича отгоре надолу.

• влизане в Short (SELL) – индикаторът MACD е над нулата, движи се отгоре надолу,

сигналната линия го пресича отдолу нагоре.

• изход от Long – при задействане на тейкпрофит, плаващ стоп или при пресичане на MACD 

със сигналната му линия (MACD е над нулата, движи се отгоре надолу, сигналната линия го 

пресича отдолу нагоре).  

• изход от Short – при задействане на тейкпрофит, плаващ стоп или при пресичане на MACD 

със сигналната му линия (MACD е под нулата, движи се отдолу нагоре, сигналната линия го 

пресича отгоре надолу). 

 

Забележка: За изключване от анализа на незначителните изменения на индикатора MACD (които 
на графиката приличат на малки 'хълмчета'), ще въведем допълнителен контрол на размера на 
изобразяваните 'хълмчета' под формата на следното условие: големината на индикатора не трябва 
да е по-малка от 5 единици от минималната цена (5*Point, което за USD/CHF е равно на 0.0005, за 
USD/JPY = 0.05). 

1 етап – създаване на описанието на експертната система 

В прозореца "Navigator" кликнете с десния бутон на мишката 
върху раздела "Expert Advisors", и в появилия се прозорец 
изберете командата "Create". Прозорецът за въвеждане на 
първоначални настройки на експертната система ще ви 
предложи да въведете следните данни: "Name" – име на 
програмата, "Author" – автор на програмата, "Link" – линк към 
уеб сайта.

2 етап – създаване на първичната структура на програмата 

Кодът на тестовата експертна система обикновено заема само няколко страници, но и този обем 
често е труден за възприемане. Особено, ако ние с вас не сме професионални програмисти... Иначе 
това описание нямаше да се налага да се пише.

За запознаване със структурата на стандартна експертна система нека погледнем следното 
описание:

1.   Инициализация на променливите 

2.   Първична проверка на данните 

• проверка на графиката, броя на баровете в графиката 

• проверка на стойностите на външните променливи Lots, S/L, T/P, T/S 

3.   Определяне на вътрешните променливи за бърз достъп до данните 

4.   Проверка на търговския терминал – празен ли е ? Ако е празен, тогава:

• проверка: има ли пари по сметката и т.н...  

• може ли да се отвори дълга позиция (BUY)? 

- Отваряне на дълга позиция и излизане 

• Може ли да се отвори къса позиция (SELL)? 

- Отваряне на къса позиция и излизане 

Излизане от експертната система...  

5.   Управление на по-рано отворените позиции в цикъла 

• Ако това е дълга позиция 

- Трябва ли тя да се затвори?

- Трябва ли да се премести плаващия стоп ?  

• Ако това е къса позиция 

- Трябва ли тя да се затвори?

- Трябва ли да се премести плаващия стоп ? 

Както виждате, тази експертна система съдържа само 4 основни блока.

Сега ще се опитаме стъпка по стъпка да напишем кода на програмата, въз основа на предложената 
по-горе структурна схема:

1.      Инициализация на променливите 

Всички променливи, които ще се използват в експертната система, трябва да бъдат описани 
предварително, в съответствие със синтаксиса на езика MetaQuotes Language 4. Затова в началото 
на програмата слагаме блок за инициализация на променливите.

2.     extern double TakeProfit = 50; 
 3.     extern double Lots = 0.1; 
 4.     extern double TrailingStop = 30; 
 5.     extern double MACDOpenLevel=3; 
 6.     extern double MACDCloseLevel=2; 
 7.     extern double MATrendPeriod=26;  

В MetaQuotes Language 4 е въведено понятието „допълнителни потребителски променливи”. Тези 
променливи да бъдат поставени отвън, без да се намесваме в изходния текст на експертната 
система. Това придава допълнителна гъвкавост на програмата. Променливата MATrendPeriod 
представлява допълнителна потребителска променлива. В началото на програмата слагаме 
описанието на тази променлива.

extern double MATrendPeriod=26; 

8.     Първична проверка на данните 

Тази част от кода на програмата обикновено може да се мести от една експертна система в друга,
без да се внасят значителни промени в нея. Това е стандартен блок за проверка:

9.            // първична проверка на данните 

10.            // трябва да се уверите, че експертната система работи с нормална графика и

11.            // потребителят правилно е определил външните променливи (Lots, StopLoss, 

12.            // TakeProfit, TrailingStop) 

13.            // в нашия случай проверяваме само TakeProfit 

14.                if(Bars<100) 

15.                 { 

16.                   Print("bars less than 100"); 

17.                   return(0);   

18.                 } 

19.               if(TakeProfit<10) 

20.                { 

21.                  Print("TakeProfit less than 10"); 

22.                  return(0);  // проверяваме TakeProfit 
23.                }  

24.    Определяне на вътрешните променливи за бърз достъп до данните 

При писане на кода на програмата достатъчно често ни се налага да работим със стойностите на 
индикаторите или с изчислените стойности.

25.        int start() 
 26.          { 
 27.           double MacdCurrent, MacdPrevious, SignalCurrent; 
 28.           double SignalPrevious, MaCurrent, MaPrevious; 
 29.           int cnt, ticket, total; 

За улесняване на кодирането и осигуряване на по-бърз достъп се използва предварително 
определяне на данните във вътрешните променливи.

MacdCurrent=iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_MAIN,0); 
 MacdPrevious=iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_MAIN,1); 
 SignalCurrent=iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_SIGNAL,0); 
 SignalPrevious=iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_SIGNAL,1); 
 MaCurrent=iMA(NULL,0,MATrendPeriod,0,MODE_EMA,PRICE_CLOSE,0); 
 MaPrevious=iMA(NULL,0,MATrendPeriod,0,MODE_EMA,PRICE_CLOSE,1); 

Сега вместо огромния запис iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_MAIN,0) в текста на 
програмата може да се използва MacdCurrent.

30.      Проверка на търговския терминал – празен ли е? Ако е празен, тогава:
В дадената експертна система ние използваме само позициите, отворени по текущата пазарна цена 
и не използваме отложени поръчки, макар че за по-голяма сигурност е добре да проверим 
търговския терминал за наличието на пласирани поръчки:

31.        // сега трябва да се определи, в какво състояние се намира търговският терминал?
32.        // да проверим, дали има по-рано отворени позиции или поръчки?
33.            total=OrdersTotal(); 

 34.            if(total<1)  
 35.              {  

• проверка: има ли пари по сметката и т.н... 

Преди анализирането на пазарната ситуация трябва да се провери състоянието на вашата 

сметка – има ли във сметката ви парични средства за отваряне на позиция?

// няма нито една отворена поръчка 
// за всеки случай да проверим, имаме ли пари по сметката?
// стойността 1000 е дадена като пример, в повечето случаи може да се отвори 1 лот 
if(AccountFreeMargin()<(1000*Lots)) 

 {
Print("We have no money. Free Margin = ", AccountFreeMargin()); 

 return(0); 
 }

• проверка на възможността за отваряне на дълга позиция (BUY) 

Условие за влизане в дълга позиция: MACD е под нулата, движи си отдолу нагоре,

сигналната линия го пресича отгоре надолу. На езика MQL 4 това се записва по следния 

начин (обърнете внимание, че се работи с определените по-рано стойности на 

индикаторите в променливите): 

 // проверка на възможността за отваряне на дълга позиция(BUY) 
 if(MacdCurrent<0 && MacdCurrent>SignalCurrent && 
 MacdPrevious<SignalPrevious && 
 MathAbs(MacdCurrent)>(MACDOpenLevel*Point) && 
 MaCurrent>MaPrevious) 
 {

ticket=OrderSend(Symbol(),OP_BUY,Lots,Ask,3,0, 
 Ask+TakeProfit*Point,"macd sample", 
 16384,0,Green); 
 if(ticket>0) 
 {

if(OrderSelect(ticket,SELECT_BY_TICKET,MODE_TRADES)) 
 Print("BUY order opened : ",OrderOpenPrice()); 
 }

else Print("Error opening BUY order : ",GetLastError()); 
 return(0); 
 }

По-горе ние споменахме за осигуряването на допълнителен контрол на размера на изобразените на 
графиката 'хълмчета'. Променливата MACDOpenLevel представлява 'потребителска' променлива,
която може да се променя, без да се внасят промени в текста на програмата, което, от своя страна,
води до повишаване на гъвкавостта на контрола. Вмъкваме описанието на тази променлива в
началото на програмата (а също и описанието на променливата, която се използва по-долу). 

• проверка на възможността за отваряне на къса позиция (SELL) 

Условие за влизане в къса позиция: MACD е над нулата, движи се отгоре надолу,

сигналната линия го пресича отдолу нагоре. Това се записва по следния начин:

// проверка на възможността за отваряне на къса позиция (SELL) 

 if(MacdCurrent>0 && MacdCurrent<SignalCurrent && 

 MacdPrevious>SignalPrevious && 

 MacdCurrent>(MACDOpenLevel*Point) && 

 MaCurrent>MaPrevious) 

 {

ticket=OrderSend(Symbol(),OP_SELL,Lots,Bid,3,0, 

 Bid-TakeProfit*Point,"macd sample", 

 16384,0,Red); 

 if(ticket>0) 

 {

if(OrderSelect(ticket,SELECT_BY_TICKET,MODE_TRADES)) 

 Print("SELL order opened : ",OrderOpenPrice()); 

 }

else Print("Error opening SELL order : ",GetLastError()); 

 return(0); 

 }

36.       // тук свършва проверката на възможността за отваряне на нови позиции.
37.       // новите позиции не са били отворени и просто излизаме 
38.       return(0); 

 39.      } 

40.      Контрол на по-рано отворените позиции в цикъла 

41.      // преминаваме към важната част на експертната система – контрола на отворените позиции 

42.      // 'важното е правилно да се влезе в пазара, но да се излезе е още по-важно...' 

43.      for(cnt=0;cnt<total;cnt++) 

44.        { 

45.         OrderSelect(cnt, SELECT_BY_POS, MODE_TRADES); 

46.         if(OrderType()<=OP_SELL &&   // това отворена позиция ли е? OP_BUY или OP_SELL  

47.            OrderSymbol()==Symbol())  // съвпада ли инструментът?

48.           { 

"Cnt" е променлива на цикъла, която трябва да бъде описана в началото на програмата по следния 
начин : 

 int cnt = 0;  

• Ако това е дълга позиция 

if(OrderType()==OP_BUY)   // отворена е дълга позиция 
{

- Трябва ли да се затвори?

Условие за излизане от дълга позиция: при пресичане на MACD със сигналната му линия, когато 
MACD е над нулата, движи се отгоре надолу, сигналната линия го пресича отдолу нагоре.

// да проверим, дали позицията трябва да се затвори вече?
if(MacdCurrent>0 && MacdCurrent<SignalCurrent && 

 MacdPrevious>SignalPrevious && 
 MacdCurrent>(MACDCloseLevel*Point)) 
 {

OrderClose(OrderTicket(),OrderLots(),Bid,3,Violet); // затваряме позицията 
return(0); // излизаме 

}

- Трябва ли да се премести плаващия стоп?

Плаващият стоп се поставя само тогава, когато позицията е вече печеливша и превишава 

големината на плаващия стоп в пипсове, както и ако новото ниво на стопа е по-добро от 
предишното.

// да проверим, може ли/трябва ли да се постави вече плаващ стоп?
if(TrailingStop>0)   

 {
if(Bid-OrderOpenPrice()>Point*TrailingStop) 

 {
if(OrderStopLoss()<Bid-Point*TrailingStop) 

 {
OrderModify(OrderTicket(),OrderOpenPrice(), 

 Bid-Point*TrailingStop,OrderTakeProfit(),0,Green); 
 return(0); 
 }

}
}

Затваряме фигурната скоба 

}

• Ако това е къса позиция 

else // иначе това е къса позиция 
{

- Трябва ли да се затвори?

Условие за излизане от къса позиция: при пресичане на MACD със сигналната му линия, когато 
MACD е под нулата, движи се отдолу нагоре, а сигналната линия го пресича отгоре надолу.

// да проверим, дали позицията трябва да се затвори вече?
if(MacdCurrent<0 && MacdCurrent>SignalCurrent && 

 MacdPrevious<SignalPrevious && 
 MathAbs(MacdCurrent)>(MACDCloseLevel*Point)) 
 {

OrderClose(OrderTicket(),OrderLots(),Ask,3,Violet); // затваряме позицията 
return(0); // излизаме 

}

- Трябва ли да се премести плаващия стоп?

Плаващият стоп се поставя само тогава, когато позицията е вече печеливша и превишава 
големината на плаващия стоп в пипсове, както и ако новото ниво на стопа е по-добро от 
предишното.

// да проверим, може ли/трябва ли да се постави вече плаващ стоп?
if(TrailingStop>0)   

 {
if((OrderOpenPrice()-Ask)>(Point*TrailingStop)) 

 {
if((OrderStopLoss()>(Ask+Point*TrailingStop)) || (OrderStopLoss()==0)) 

 {
OrderModify(OrderTicket(),OrderOpenPrice(), 

 Ask+Point*TrailingStop,OrderTakeProfit(),0,Red); 
 return(0); 
 }

}
}

Затваряме всички останали фигурни скоби 

}
}

}
return(0); 

 }

Ето така, стъпка по стъпка се пише експертна система...  

 
3 етап – сглобяване на крайния код на програмата 

Да отворим прозореца за задаване на настройките на експертната система (това става с помощта на 
бутона F7 или "Properties..." в менюто). На екрана ще се появи прозорец за задаване на външните 
настройки на параметрите, необходими за осъществяването на по-нататъшна работа:

Нека сглобим сега целия код на програмата, описан на части в предния раздел... 

//+------------------------------------------------------------------+ 

//|                                                  MACD Sample.mq4 | 

//|                      Copyright й 2005, MetaQuotes Software Corp. | 

//|                                       http://www.metaquotes.net/ | 

//+------------------------------------------------------------------+ 

extern double TakeProfit = 50; 

extern double Lots = 0.1; 

extern double TrailingStop = 30; 

extern double MACDOpenLevel=3; 

extern double MACDCloseLevel=2; 

extern double MATrendPeriod=26; 

//+------------------------------------------------------------------+ 

//|                                                                  | 

//+------------------------------------------------------------------+ 

int start() 

{

double MacdCurrent, MacdPrevious, SignalCurrent; 

double SignalPrevious, MaCurrent, MaPrevious; 

int cnt=0, ticket, total; 

// първична проверка на данните 

// трябва да се уверите, че експертната система работи с нормална графика и

// потребителят е определил правилно външните променливи (Lots, StopLoss, 

// TakeProfit, TrailingStop) 

// в нашия случай проверяваме само TakeProfit 

if(Bars<100) 

{

Print("bars less than 100"); 

return(0);   

}

if(TakeProfit<10) 

{

Print("TakeProfit less than 10"); 

return(0);  // проверяваме TakeProfit 

}

// за улесняване на кода на програмата и за осигуряване на по-бърз достъп, да зададем 
стойностите на временните променливи 

// с помощта на съответните данни на индикаторите 

MacdCurrent=iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_MAIN,0); 

MacdPrevious=iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_MAIN,1); 

SignalCurrent=iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_SIGNAL,0); 

SignalPrevious=iMACD(NULL,0,12,26,9,PRICE_CLOSE,MODE_SIGNAL,1); 

MaCurrent=iMA(NULL,0,MATrendPeriod,0,MODE_EMA,PRICE_CLOSE,0); 

MaPrevious=iMA(NULL,0,MATrendPeriod,0,MODE_EMA,PRICE_CLOSE,1); 

total=OrdersTotal(); 

if(total<1)  

{

// няма нито една отворена поръчка 

if(AccountFreeMargin()<(1000*Lots)) 

{

Print("We have no money. Free Margin = ", AccountFreeMargin()); 

return(0);   

}

// проверка на възможността за отваряне на дълга позиция (BUY) 

if(MacdCurrent<0 && MacdCurrent>SignalCurrent && MacdPrevious<SignalPrevious && 

MathAbs(MacdCurrent)>(MACDOpenLevel*Point) && MaCurrent>MaPrevious) 

{

ticket=OrderSend(Symbol(),OP_BUY,Lots,Ask,3,0,Ask+TakeProfit*Point, 

"macd sample",16384,0,Green); 

if(ticket>0) 

{

if(OrderSelect(ticket,SELECT_BY_TICKET,MODE_TRADES)) 

Print("BUY order opened : ",OrderOpenPrice()); 

}

else Print("Error opening BUY order : ",GetLastError());  

return(0);  

}

// проверка на възможността за отваряне на къса позиция (SELL) 

if(MacdCurrent>0 && MacdCurrent<SignalCurrent && MacdPrevious>SignalPrevious &&  

MacdCurrent>(MACDOpenLevel*Point) && MaCurrent>MaPrevious) 

{

ticket=OrderSend(Symbol(),OP_SELL,Lots,Bid,3,0,Bid-TakeProfit*Point, 

"macd sample",16384,0,Red); 

if(ticket>0) 

{

if(OrderSelect(ticket,SELECT_BY_TICKET,MODE_TRADES)) 

Print("SELL order opened : ",OrderOpenPrice()); 

}

else Print("Error opening SELL order : ",GetLastError());  

return(0);  

}

return(0); 

}

// преминаваме към важната част на експертната система – контрола на отворените позиции 

// 'важното е правилно да се влезе в пазара, но да се излезе е още по-важно...' 

for(cnt=0;cnt<total;cnt++) 

{

OrderSelect(cnt, SELECT_BY_POS, MODE_TRADES); 

if(OrderType()<=OP_SELL &&   // това отворена позиция ли е? OP_BUY или OP_SELL  

OrderSymbol()==Symbol())  // съвпада ли инструментът?

{

if(OrderType()==OP_BUY)   // отворена е дълга позиция 

{

// да проверим, дали позицията трябва да се затвори вече?

if(MacdCurrent>0 && MacdCurrent<SignalCurrent && MacdPrevious>SignalPrevious && 

MacdCurrent>(MACDCloseLevel*Point)) 

{

OrderClose(OrderTicket(),OrderLots(),Bid,3,Violet); // затваряме позицията 

return(0); // излизаме 

}

// да проверим, дали вече може/трябва да се постави плаващ стоп?

if(TrailingStop>0)   

{

if(Bid-OrderOpenPrice()>Point*TrailingStop) 

{

if(OrderStopLoss()<Bid-Point*TrailingStop) 

{

OrderModify(OrderTicket(),OrderOpenPrice(),Bid-Point*TrailingStop, 

OrderTakeProfit(),0,Green); 

return(0); 

}

}

}

}

else // иначе това е къса позиция 

{

// да проверим, дали позицията трябва да се затвори вече?

if(MacdCurrent<0 && MacdCurrent>SignalCurrent && 

MacdPrevious<SignalPrevious && MathAbs(MacdCurrent)>(MACDCloseLevel*Point)) 

{

OrderClose(OrderTicket(),OrderLots(),Ask,3,Violet); // затваряме позицията 

return(0); // излизаме 

}

// да проверим, дали вече може/трябва да се постави плаващ стоп?

if(TrailingStop>0)   

{

if((OrderOpenPrice()-Ask)>(Point*TrailingStop)) 

{

if((OrderStopLoss()>(Ask+Point*TrailingStop)) || (OrderStopLoss()==0)) 

{

OrderModify(OrderTicket(),OrderOpenPrice(),Ask+Point*TrailingStop, 

OrderTakeProfit(),0,Red); 

return(0); 

}

}

}

}

}

}

return(0); 

}

// край.

В края на задаване на настройките на експертната система трябва да се определят стойностите на 
външните променливи "Lots = 1", "Stop Loss (S/L) = 0" (не се използва), "Take Profit (T/P) = 120" 
(подходяща за часови графики), "Trailing Stop (T/S) = 30". Разбира се, вие можете да определите 
свои стойности.
Натиснете бутона "Verify" и, ако не са намерени грешки (между другото, вие можете да копирате 
текста на предложената по-горе програма направо в редактора MetaEditor), натиснете бутона 
"Save", за да се запише създадената от вас експертна система.


